
Abstract. In this short note we prove the unirationality of Hur-
witz spaces of 6-gonal curves of genus g with 5 ≤ g ≤ 28 or
g = 30, 31, 35, 36, 40, 45. Key ingredient is a liaison construction in
P1 × P2. By semicontinuity, the proof of the dominance of this con-
struction is reduced to a computation of a single curve over a finite
field.

1 Introduction

The study of the birational geometry of moduli spaces of curves with additional
structures such as marked points or line bundles is a central topic in algebraic
geometry, see for example the books [HM98] and [ACG11]. The Hurwitz space
H (d, w) parametrizes d-sheeted branched simple covers of the projective line
by smooth curves of genus g with branch divisor of degree w = 2g + 2d− 2 up
to isomorphism,

H (d, 2g+2d− 2) = {C d:1−−→ P1 simply branched | C smooth of genus g}/ ∼ .

It is a classical result by Arbarello and Cornalba [AC81] based on a work of
Segre [Seg28] that theses spaces are unirational for all d ≤ 5 and all g ≥ d− 1
and in few cases for higher gonality, namely for d = 6 and 5 ≤ g ≤ 10 or g = 12
and for d = 7 and g = 7.
In this paper we present the following extension of this result to significantly
higher genus for 6-gonal curves.

Theorem 1.1. Over an algebraically closed field of characteristic zero, the
Hurwitz spaces H (6, 2g + 10) of 6-gonal curves of genus g are unirational for

5 ≤ g ≤ 28 or g = 30, 31, 33, 35, 36, 40, 45. (1)

Our proof is based on the observation that a general 6-gonal curve in P1 × P2

can be linked in two steps to the union of a rational curve and a collection of
lines. It turns out that for small genera this process can be reversed by starting
with a general rational curve and general lines.
To show that the described construction yields a parametrization of the Hurwitz
space, we only need to run the construction for a single curve over a finite field.
Semicontinuity then ensures that all assumptions we made actually hold for an
open dense subset of H (6, 2g+10) in characteristic zero. Since the construction
works a priori only for finitely many genera we settle for a computer-aided
verification using the computer algebra system Macaulay2 [GS].
An immediate consequence of our approach is that in the considered cases the
general 6-gonal curve has a plane model as expected from Brill-Noether theory.

Corollary 1.2. For g among (1) and d =
⌈

2
3g + 2

⌉

the Brill-Noether locus
W 2

d (C) of a general curve C ∈ H (6, 10+ 2g) has a smooth generically reduced
component of expected dimension ρ(g, 2, d).

2 Preliminaries

Throughout this paper, we fix the following notation: Let P = P1 × P2 be
the product of the projective line and the projective plane over a field K with
projections π1 : P → P1 and π2 : P → P2. For a, b ∈ Z we write

OP(a, b) = OP1(a)⊠ OP2(b) = π∗
1OP1(a)⊗ π∗

2OP2(b)

and denote with R =
⊕

i,j H
0(P,OP(i, j)) ∼= K[x0, x1, y0, y1, y2] the bihomoge-

neous coordinate ring of P. By a curve C in P, we mean an equidimensional
subscheme of codimension 2 which is locally a complete intersection. We say
that C is (geometrically) linked to a curve C′ ⊂ P by a complete intersection
X ⊂ P if C and C′ have no common components and C ∪ C′ = X . The Chow
ring of P is generated by classes α and β which are the pullback of a point in
P1 and the pullback of a hyperplane in P2, respectively. The bidegree (d1, d2)
of a curve C is given by d1 = [C].α and d2 = [C].β.

As in the classical setting of liaison of subschemes in Pn, we have the following

Proposition 2.1 (Exact sequence of liaison). Let C be a curve of bidegree
(d1, d2) that is linked to C′ via a complete intersection X defined by forms of
bidegree (a1, b1) and (a2, b2). We set a = a1 + a2 and b = b1 + b2.

(a) There is an exact sequence

0 → ωC → ωX → OC′(a− 2, b− 3) → 0.

(b) The curve C′ has bidegree (d′1, d
′
2) = (b1b2 − d1, a1b2 + a2b1 − d2) and

arithmetic genus pa(C
′) = pa(X)−(d1(a− 2) + d2(b− 3) + (1− pa(C))).

Proof. To prove the first part, consider the standard exact sequence

0 → IC/X → OX → OC → 0

and apply HomOP
(−, ωP). From the long exact sequence, we get

0 → ωC → ωX → Ext2(IC/X , ωP) → 0

but Ext2(IC/X , ωP) = OC′(a − 2, b − 3) since C and C′ are linked by X . The
formula for the genus follows immediately. To compute the bidegree, note that
[C] + [C′] = [X ] = (b1b2)β

2 + (a1b2 + a2b1)αβ in the Chow ring of P.

Recall the following well-known fact about minimal resolutions of points in the
plane.

Proposition 2.2. Let ∆ be a collection of δ general points in P2 and let k
be maximal under the condition ε = δ −

(

k+1
2

)

≥ 0. Then the minimal free
resolution of O∆ is of the form

0 → G → F → OP2 → O∆ → 0

with F = O(−k)k+1−ε and G = O(−k − 1)k−2ε ⊕ O(−k − 2)ε if 2ε ≤ k and
F = O(−k)k+1−ε ⊕ O(−k − 1)2ε−k and G = O(−k − 2)ε else.

Proof. [Gae51]

We also note the following simple but useful criterion for the irreducibility of
plane curves. Recall that a variety over a field K is called absolutely irreducible
if it is irreducible as a variety over the algebraic closure K.

Proposition 2.3. Let C be a plane curve of degree d with δ ≤ d(d−3)
2 ordinary

double points and no other singularities. If the singular locus ∆ of C has a
resolution as in 2.2 then C is absolutely irreducible.

Proof. Assume that C decomposes into two curves C1 and C2 of degree d1 and
d2 defined by homogeneous polynomials f1 and f2. By assumption, C1 and C2

intersect transversely in d1 · d2 distinct points. First, we reduce to the case
d1, d2 ≤ k where k =

⌈

(
√
9 + 8δ − 3)/2

⌉

is the minimal degree of generators of

I∆. Clearly, the case that one of the generators has degree strictly larger than
k+1 is not possible since I∆ ⊂ (f1, f2) is generated in degree k and (possibly)
k+1. The cases d1 = k+1, say, and d2 ≤ k+1 can be excluded by considering
the number of minimal generators of I∆ in degrees k and k + 1.
We are left with the case d1, d2 ≤ k. Trivially, we can assume that δ−d1d2 ≥ 0.
A polynomial of the form sf1 + tf2 of degree k lies in I∆ if it vanishes at the
remaining δ − d1d2 points. Hence,

h0(I∆(k)) ≥
(

k − d1 + 2

2

)

+

(

k − d2 + 2

2

)

− δ + d1d2

= 2

(

k + 2

2

)

+

(

d− 1

2

)

− (dk + 1)− δ

But this is strictly larger than
(

k+2
2

)

− δ since d ≥ k + 3.

Recall from [ACGH85] the following facts from Brill-Noether theory: For a
fixed smooth curve C of genus g, the Brill-Noether locus

W r
d (C) = {L ∈ Picd(C) | h0(L) ≥ r + 1} (2)

is of dimension at least equal to the Brill-Noether number

ρ(g, r, d) = g − (r + 1)(g − d+ r). (3)

The tangent space at a linear series L ∈ W r
d (C) rW r+1

d (C) is the dual of the
cokernel of the Petri-map

µL : H0(C,L)⊗H0(C, ωC ⊗ L−1) → H0(C, ωC) (4)

Hence, W r
d (C) is smooth of dimension ρ at L if and only if µL is injective.

Proposition 2.4. Let C be a smooth curve of genus g ≥ 3 with |D| a base
point free g

2
d, d =

⌈

2g
3 + 2

⌉

, such that the image of C under the associated

map is a plane curve with δ =
(

d−1
2

)

− g ordinary double points and no other
singularities. If the singular locus ∆ has a resolution as in 2.2 then |D| is a
smooth point in W 2

d (C).

Proof. By adjunction, the Petri map for O(D) can be identified with

H0(P2,O(1))⊗H0(P2,I∆(d− 4)) → H0(P2,I∆(d− 3)).

Under the given assumptions the minimal degree of generators of I∆ is precisely
k = d − 4. As 2ε ≤ k we see from the minimal free resolution of I∆ that the
Petri map is injective since there are no linear relations among the generators
of degree k and k + 1.

3 Liaison construcion

For g ≥ 5, let f : C → P1 be an element of H (6, 10 + 2g) and let O(D1) =
f∗OP1(1) be the 6-gonal bundle. We assume that C has a line bundle O(D2)
such that |D2| is a complete base point free g

2
d with d = d(g) =

⌈

2g
3 + 2

⌉

minimal under the condition that the Brill-Noether number ρ(g, 2, d) ≥ 0.
Suppose further that the map

ϕ : C
|D1|,|D2|−−−−−−→ PH0(O(D1))× PH0(O(D2)) = P (5)

is an embedding. In particular, this is the case when we assume that the plane
model has only ordinary double points and no other singularities and for any
node p the points in the preimage of p under C → P2 are not identified under
the map to P1.
Hence, we can identify C with its image under ϕ. Furthermore, we assume
that the map H0(OP(a, 3)) → H0(OC(a, 3)) is of maximal rank for all a ≥ 1.
To simplify matters, assume g ≡ 0 (12) for the moment. By the maximal rank
assumption, we have

aCubic := min{a | H0(IC(a, 3)) 6= 0} =
g

4
(6)

and h0(IC(aCubic, 3)) = 3. Let X = V (f1, f2) be the complete intersection
defined by two general sections fi ∈ H0(IC(ai, bi)) of bidegrees (a1, b1) =
(a2, b2) = (aCubic, 3). The curve C′, obtained by liaison of C by X , is smooth
of bidegree (3, 56g − 2) and genus g′ = g

2 − 3 with h0(IC′(aCubic, 3)) ≥ 2.
The geometric situation is understood best when thinking of C as a family of
collections of plane points over P1. We expect the general fiber of C to be a
collection of 6 points in P2 which are cut out by 4 cubics. We expect a finite
number ℓ of distinguished fibers where the points lie on a conic as this is a
codimension 1 condition on the points. Since the residual three points under
liaison are collinear exactly in the distinguished fibers we can compute ℓ by
examining the geometry of C′. The projection of C′ to P2 yields a divisor D′

2

of degree d′ > g′ + 2. Our claim is that ℓ = d′ − (g′ + 2). Indeed, the image of
C′ under the associated map

ψ : C′ → P1 × PH0(C′,O(D′
2)) = P1 × Pd′−g′

(7)

lies on the graph of the projection S → P1 where S is a 3-dimensional scroll of
degree d′ − g′ − 2 swept out by the 3-gonal series |D′

1|, i.e.

ψ(C′) ⊂ P1 × S =
⋃

D∈|D′

1
|

{D} ×D. (8)

See [Sch86] for a proof of this fact. C′ is obtained from ψ(C′) by projection
from a linear subspace P1×V ⊂ P1×Pd′−g′

of codimension 3. A general space
V intersects S in precisely d′ − g′ − 2 points lying in distinct fibers over P1.

Documenta Mathematica 17 (2012) 627–640



632 Florian Geiß

Clearly, under the projection the points of D ∈ |D′
1| are mapped to 3 collinear

points if and only if V meets the corresponding fiber of S.
To keep things neat, we consider again the case g ≡ 0 (12) which implies
ℓ = 1

3g − 1. Suppose further that ℓ ≡ 1 (3). If we assume that the map
H0(OP(a, 2)) → H0(OC′(a, 2)) is of maximal rank for all a ≥ 1 then

aConic = min{a | H0(IC′(a, 2)) 6= 0} =
g′ + 2ℓ+ 1

3
(9)

and h0(IC′(aConic, 2)) = 2. Let X ′ = V (f ′
1, f

′
2) be defined by two general

forms f ′
i ∈ H0(I (a′i, b

′
i)) of bidegrees (a′1, b

′
1) = (a′2, b

′
2) = (aConic, 2) and let

C′′ denote the curve that is linked to C′ via X ′. The general fiber of C′′

consists of a single point. In a distinguished fiber the conics of the complete
intersection are reducible and have the line spanned by the points of the fiber
of C′ as a common factor. Hence, C′′ is a rational curve together with ℓ lines.
The rational curve has degree

d′′ =
g′ + 2ℓ− 2

3
=

7

18
g − 7

3
. (10)

Turning things around we see that the difficulty lies in reversing the first linkage
step. Indeed, a simple counting argument shows that for any g, the union C′′ of
ℓ general lines in P and the graph of a general rational normal curve of degree
d′′ we have

min{a ∈ Z| H0(IC′′(a, 2)) 6= 0} =

⌈

2d′′ + 3ℓ

5

⌉

− 1 ≤ aConics.

Hence, we always obtain a trigonal curve C′ as desired. However, for
general choices of C′′ and X ′ we expect that the map H0(OP(a, 3)) →
H0(OC′(aCubic, 3)) is of maximal rank. In the case g ≡ 0(12), this restriction
yields h0(IC′(aCubic, 3)) = − g

4 + 12, hence g < 48. Checking all congruency
classes of g, we expect that C′ can be linked to a general curve C exactly in
the cases

5 ≤ g ≤ 28 or g = 30, 31, 33, 35, 36, 40, 45. (11)

Table 1 lists the appearing numbers for all values of g in (11).
Summarizing, we obtain for g among (11) the following unirational construction
for curves in H (6, 10 + 2g):

1. We start with a general rational curve of degree d′′ in P together with a
collection of ℓ general lines. Call the union C′′.

2. We choose two general forms f ′
i ∈ H0(IC′′(a′i, b

′
i)), i = 1, 2, that define

a complete intersection X ′ and obtain a trigonal curve C′ = X ′ r C′′ of
degree d′ and genus g′.

3. We choose two general forms fi ∈ H0(IC′(ai, bi)), i = 1, 2, that define a
complete intersection X and obtain a 6-gonal curve C = X r C′.

It remains to show that the construction actually yields a parametrization of
the Hurwitz spaces.

g d (a1, b1), (a2, b2) g′ d′ (a′1, b
′
1), (a

′
2, b

′
2) ℓ d′′

5 6 (2, 3), (2, 3) 2 6 (3, 2), (2, 2) 2 2
6 6 (2, 3), (1, 3) 0 3 (1, 2), (1, 2) 1 0
7 7 (2, 3), (2, 3) 1 5 (2, 2), (2, 2) 2 1
8 8 (3, 3), (2, 3) 2 7 (3, 2), (3, 2) 3 2
9 8 (2, 3), (2, 3) 0 4 (2, 2), (2, 2) 2 2
10 9 (3, 3), (3, 3) 4 9 (4, 2), (4, 2) 3 4
11 10 (3, 3), (3, 3) 2 8 (4, 2), (4, 2) 4 4
12 10 (3, 3), (3, 3) 3 8 (4, 2), (3, 2) 3 3
13 11 (4, 3), (3, 3) 4 10 (5, 2), (4, 2) 4 4
14 12 (4, 3), (4, 3) 5 12 (6, 2), (5, 2) 5 5
15 12 (4, 3), (4, 3) 6 12 (5, 2), (5, 2) 4 4
16 13 (4, 3), (4, 3) 4 11 (5, 2), (5, 2) 5 4
17 14 (5, 3), (5, 3) 8 16 (7, 2), (7, 2) 6 6
18 14 (5, 3), (4, 3) 6 13 (6, 2), (6, 2) 5 6
19 15 (5, 3), (5, 3) 7 15 (7, 2), (7, 2) 6 7
20 16 (6, 3), (5, 3) 8 17 (8, 2), (8, 2) 7 8
21 16 (5, 3), (5, 3) 6 14 (7, 2), (6, 2) 6 6
22 17 (6, 3), (6, 3) 10 19 (9, 2), (8, 2) 7 8
23 18 (6, 3), (6, 3) 8 18 (9, 2), (8, 2) 8 8
24 18 (6, 3), (6, 3) 9 18 (8, 2), (8, 2) 7 7
25 19 (7, 3), (6, 3) 10 20 (9, 2), (9, 2) 8 8
26 20 (7, 3), (7, 3) 11 22 (10, 2), (10, 2) 9 9
27 20 (7, 3), (7, 3) 12 22 (10, 2), (10, 2) 8 10
28 21 (7, 3), (7, 3) 10 21 (10, 2), (10, 2) 9 10
30 22 (8, 3), (7, 3) 12 23 (11, 2), (10, 2) 9 10
31 23 (8, 3), (8, 3) 13 25 (12, 2), (11, 2) 10 11
33 24 (8, 3), (8, 3) 12 24 (11, 2), (11, 2) 10 10
35 26 (9, 3), (9, 3) 14 28 (13, 2), (13, 2) 12 12
36 26 (9, 3), (9, 3) 15 28 (13, 2), (13, 2) 11 13
40 29 (10, 3), (10, 3) 16 31 (15, 2), (14, 2) 13 14
45 32 (11, 3), (11, 3) 18 34 (16, 2), (16, 2) 14 16

Table 1: Numerical data for all cases of the construction

4 Proof of the dominance

Theorem 4.1. For all (g, d) as in Table 1, there is a unirational component Hg

of the Hilbert scheme Hilb(6,d),g(P) of curves in P of bidegree (6, d) and genus
g. The generic point of Hg corresponds to a smooth absolutely irreducible curve
C such that the map H0(OP(a, 3)) → H0(OC(a, 3)) is of maximal for all a > 1.

Proof. The crucial part is to prove the existence of a curve with the desired
properties. Code 5.1 implements the construction above for any given value of
g in (11) and establishes the existence of a smooth and absolutely irreducible
curve Cp of given genus and bidegree defined over a prime field Fp. This com-
putation can be regarded as the reduction of a computation over Q which yields
some curve C0. This curve is already defined over the rationals, since all con-
struction steps invoke only Groebner basis computations. By semicontinuity,
C0 is also smooth, absolutely irreducible and of maximal rank.
Again, by semicontinuity, there is a Zariski open neighborhood U ⊂
Hilb(6,d),g(P) of points corresponding to smooth absolutely irreducible curves
that fulfill the maximal rank condition. Let AN be the parameter-space for
all the choices made in the construction, i.e. the space of coefficients of the
polynomials defining C′′ and the complete intersections X and X ′. The con-
struction then translates to a rational map AN

99K U defined over Q and we
set Hg to be the closure of the image of this map.

Remark 4.2. We want to point out two issues concerning the computational
verification:

1. The restriction to finite fields in the Macaulay2 computation in the ap-
pendix is only due to limitations in computational power. For very small
values of g, i.e. g ≤ 15, it is still possible to compute examples over the
rationals if all coefficients are chosen among integers of small absolute
value.

2. The reduction of C0 modulo p gives a curve Cp with desired properties for
p in an open part of Spec(Z). Hence, the main theorem is also true in
almost all characteristics p. One way to extend it to all prime numbers
would be to keep trace of all denominators in a computation over the
rationals and check case by case the primes where a bad reduction happens.
This is computationally also out of reach at the moment.

It remains to show that there exists a dominant rational map from Hg to the
Hurwitz-scheme.

Theorem 4.3. For g among (11) and Hg as in Theorem 4.1 there is a dominant
rational map

Hd 99K H (6, 10 + 2g).

Proof. Using Code 5.1 again, we check for any given value of g in (11) the
existence of a point in Hg corresponding to a smooth absolutely irreducible

curve C ⊂ P such that the projection onto P1 is simply branched and the
bundle L2 = ϕ∗OP(0, 1) is a smooth point in the corresponding W 2

d (C). By
semicontinuity, the loci of curves with this property is open and dense in Hg.
Hence, we have a rational map Hg 99K H (6, 10 + 2g). The locus of curves in
H (6, 10+2g) having a smooth component of the Brill-Noether loci of expected
dimension is also open and contains the image of [C] under this map. Since
H (6, 10 + 2g) is irreducible this locus is dense. This proves the theorem.

We want to emphasize the last statement in the proof:

Corollary 4.4. For g among (11) and d =
⌈

2
3g + 2

⌉

the Brill-Noether locus
W 2

d (C) of a general curve C ∈ H (6, 10+ 2g) has a smooth generically reduced
component of expected dimension ρ.

5 Computational Verification

The following Code for Macaulay2 [GS] realizes the unirational construction of
a 6-gonal curve of genus g as in (11) over a finite field K = Fp with random
choices for all parameters.
In order to explain the single steps in the computation, we also print the most
relevant parts of the output for the example case g = 24.

Code 5.1. We start with the following initialization:

i1 : Fp=ZZ/32009; -- a finite field

S=Fp[x_0,x1,y_0..y_2,Degrees=>{2:{1,0},3:{0,1}}];

-- Cox-ring of P^1 x P^2

m=ideal basis({1,1},S);

-- irrelevant ideal

setRandomSeed("HurwitzSpaces");

-- initialization of the random number generator

The following functions handle the numerics of the construction:

i2 : expHilbFuncIdealSheaf=(g,d,a)->

max(0,(a_0+1)*(a_1+2)*(a_1+1)/2-(a_0*d_0+a_1*d_1+1-g))

-- expected number of sections of the ideal sheaf

linkedGenus=(g,d,F,G)->(

pX:=binomial(F_0+G_0-1,1)*binomial(F_1+G_1-1,2)-

(F_0-1)*binomial(F_1-1,2)-(G_0-1)*binomial(G_1-1,2);

-- genus of the complete intersection

pX-d_0*(F_0+G_0-2)-d_1*(F_1+F_1-3)-1+g)

-- genus of the linked curve

linkedDegree=(g,d,F,G)->{F_1*G_1-d_0,F_0*G_1+G_0*F_1-d_1}

-- bidegree of the linked curve

The first step is to determine the degree d′′ of the rational curve and the number
of lines ℓ. We start by computing the bidegrees of the forms that define the
complete intersection for the linkage to the trigonal curve:

i3 : g=24;

d={6,ceiling(-g/3+g+2)};

-- choose the second degree Brill-Noether general

a=for i from 0 do

if expHilbFuncIdealSheaf(g,d,{i,3})!=0 then break i;

-- find the minimal value a s.t. H^0(IC(a,3)) nonzero

if expHilbFuncIdealSheaf(g,d,{a,3})==1 then

fX={{a+1,3},{a,3}} else fX={{a,3},{a,3}};

-- choose bidegrees of forms for the complete intersection

(d,fX)

o3 = ({6, 18}, {{6, 3}, {6, 3}})

The genus and degree of the trigonal curve and the number of lines:

i4 : g’=linkedGenus(g,d,fX_0,fX_1);

d’=linkedDegree(g,d,fX_0,fX_1);

l=d’_1-g’-2;

(g’,d’,l)

o4 = (9,{3,18},7)

We compute the bidegrees for the complete intersection for the linkage to the
rational curve

i5 : b=for i from 0 do

if expHilbFuncIdealSheaf(g’,d’,{i,2})!=0 then break i;

if expHilbFuncIdealSheaf(g’,d’,{b,2})==1 then

fX’={{b+1,2},{b,2}} else fX’={{b,2},{b,2}};

d’’=linkedDegree(g’+2*l,d’+{0,l},fX’_0,fX’_1);

dRat={{ceiling(d’’_1/2),1},{floor(d’’_1/2),1}};

(fX’,d’’)

o5 = ({{8, 2}, {8, 2}}, {1, 7})

The second step is the actual construction: First, we choose a rational curve
and random lines and compute the saturated vanishing ideal IC′′ of their union:

i6 : ICrat=saturate(ideal random(S^1,S^(-dRat)),m);

ILines=apply(l,i->ideal random(S^1,S^{{-1,0},{0,-1}}));

time IC’’=saturate(intersect(ILines|{ICrat}),ideal(x_0*y_0));

-- used 1.29537 seconds

Next, we choose random forms in IC′′ of degree b (resp. of b + 1) that define
the complete intersection X ′ and compute the saturated vanishing ideal IC′ of
the trigonal curve C′.

i7 : IX’=ideal(gens IC’’ * random(source gens IC’’,S^(-fX’)));

IC’=IX’:ICrat;

time scan(l,i->IC’=IC’:ILines_i);

time IC’sat=saturate(IC’,ideal(x_0*y_0));

-- used 2.06236 seconds

-- used 23.7319 seconds

In the final step, we compute the vanishing ideal of the 6-gonal curve C by
linking C′ with a complete intersection X given by random forms in IC′ of
degree a (resp. a+ 1).

i8 : IX=ideal(gens IC’sat * random(source gens IC’sat,S^(-fX)));

time IC=IX:IC’;

time ICsat=saturate(IC,ideal(x_0*y_0));

-- used 15.7815 seconds

-- used 3.84807 seconds

We check that C is of maximal rank in the degrees (a, 3) by looking at the
minimal generators of the saturated vanishing ideal:

i9 : tally degrees ideal mingens gb ICsat

o9 = Tally{{0, 18} => 1}

{1, 14} => 5

{1, 15} => 4

{2, 8} => 2

{2, 9} => 8

{3, 6} => 9

{4, 4} => 2

{4, 5} => 8

{5, 4} => 7

{6, 3} => 3

{7, 3} => 1

In order to check irreducibility, we compute the plane model Γ of C:

i10 : Sel=Fp[x_0,x_1,y_0..y_2,MonomialOrder=>Eliminate 2];

-- eliminination order

R=Fp[y_0..y_2]; -- coordinate ring of P^2

IGammaC=sub(ideal selectInSubring(1,gens gb sub(ICsat,Sel)),R);

-- ideal of the plane model

We check that Γ is a curve of desired degree and genus and its singular locus
∆ consists only of ordinary double points:

i11 : distinctPoints=(J)->(

singJ:=minors(2,jacobian J)+J;

codim singJ==3)

i12 : IDelta=ideal jacobian IGammaC + IGammaC; -- singular locus

distinctPoints(IDelta)

o12 = true

i13 : delta=degree IDelta;

dGamma=degree IGammaC;

gGamma=binomial(dGamma-1,2)-delta;

(dGamma,gGamma)==(d_1,g)

o13 = true

We compute the free resolution of I∆:

i14 : time IDelta=saturate IDelta;

betti res IDelta

-- used 55.063 seconds

0 1 2

o14 = total: 1 8 7

0: 1 . .

1: . . .

2: . . .

3: . . .

4: . . .

5: . . .

6: . . .

7: . . .

8: . . .

9: . . .

10: . . .

11: . . .

12: . . .

13: . 8 .

14: . . 7

This is the resolution as expected. Hence, C is absolutely irreducible by Propo-
sition 2.2 and O(D2) is a smooth point of the Brill-Noether loci by Proposition
2.4.
It remains to verify that C is actually smooth and simply branched. We com-
pute the vanishing ideal IB ⊂ K[x0, x1] of the locus B in P1 of points with
non-reduced fiber.

i15 : gensICsat=flatten entries mingens ICsat;

Icubics=ideal select(gensICsat,f->(degree f)_1==3);

-- select the cubic forms

Jacobian=diff(matrix{{y_0}..{y_2}},gens Icubics);

-- compute the jacobian w.r.t. to vars of P^2

IGraphB=minors(2,Jacobian)+Icubics;

time IGraphBsat=saturate(IGraphB,ideal(x_0*y_0));

-- used 60.2963 seconds

We check that the fibers over B are disjoint from the preimages of the double
points of the plane model. This shows that C is smooth:

i16 : time ISing=saturate(sub(IDelta,S)+IGraphBsat,ideal(S_0*S_2));

degree ISing==0

o16 = true

Finally, we verify that B is reduced of expected degree 2g + 10 and hence that
C is simply branched.

i17 : time IGraphBsat=saturate(IGraphB,ideal(x_0*y_0));

gensIGraphBsat=flatten entries mingens IGraphBsat;

IB=ideal select(gensIGraphBsat,f->(degree f)_1==0);

degree radical IB==2*g+10

o17 = true

This code is available in form of a Macaulay2 -file from [G11] for download. It
takes approximately 5 hours CPU-time on a 2.4 GHz processor to check all
cases.

du plan. C. R. Acad. Sci. Paris, 233:912–913, 1951.

[G11] Florian Geiß. Macaulay2 code for the proof of unirationality of
6-gonal curves of small genus, 2011. Available at the web page
http://www.math.uni-sb.de/ag/schreyer/home/6gonal.m2.

[GS] Daniel R. Grayson and Michael E. Stillman. Macaulay2, a software
system for research in algebraic geometry. Available at the web page
http://www.math.uiuc.edu/Macaulay2/.

Documenta Mathematica 17 (2012) 627–640

http://www.math.uni-sb.de/ag/schreyer/home/6gonal.m2
http://www.math.uiuc.edu/Macaulay2/


640 Florian Geiß

[HM98] Joe Harris and Ian Morrison. Moduli of curves, volume 187 of Graduate
Texts in Mathematics. Springer-Verlag, New York, 1998.

[MS86] G. Martens and F.-O. Schreyer. Line bundles and syzygies of trigonal
curves. Abh. Math. Sem. Univ. Hamburg, 56:169–189, 1986.

[Sch86] Frank-Olaf Schreyer. Syzygies of canonical curves and special linear
series. Math. Ann., 275(1):105–137, 1986.

[Seg28] Beniamino Segre. Sui moduli delle curve poligonali, e sopra un com-
plemento al teorema di esistenza di Reimann. Math. Ann., 100(1):537–551,
1928.

Florian Geiß
Mathematik und Informatik
Universität des Saarlandes,
Campus E2 4,
66123 Saarbrücken,
Germany
fg@math.uni-sb.de

Documenta Mathematica 17 (2012) 627–640


	Introduction
	Preliminaries
	Liaison construcion
	Proof of the dominance
	Computational Verification

